

Derivative Estimates by Undetermined Coefficients

Mth 351 April 6 2001

Bent E. Petersen

Filename: 351s2001_derivs_by_undeter_coef.mws

```
> restart;
```

In this worksheet we use Maple to obtain a number of derivative estimators. These estimators are useful for "differentiating" data obtained experimentally as a table of numbers. Symmetric estimators usually perform best, but near the beginning and near the end of the table it will be necessary to use nonsymmetric estimators.

Numeric differentiation is not a very stable process. High order estimates (and high order derivatives) tend to suffer from loss of significance errors.

We begin by looking at some symmetric 3-point formulae

```
> g:=h->A1*f(a+h)+A2*f(a)+A3*f(a-h);
```

$$g := h \rightarrow A1 f(a+h) + A2 f(a) + A3 f(a-h)$$

Since we have 3 coefficients we can specify the terms in the Taylor expansion of g up through order 2.

```
> ex:=taylor(g(h),h=0,3);
```

$$ex := (A1 f(a) + A3 f(a) + A2 f(a)) + (-A3 D(f)(a) + A1 D(f)(a)) h +$$

$$\left(\frac{1}{2} A3 (D^{(2)})(f)(a) + \frac{1}{2} A1 (D^{(2)})(f)(a) \right) h^2 + O(h^3)$$

Let's pick out the coefficients of powers of h

```
> for k from 0 to 2 do A[k]:=coeff(ex,h,k); od;
```

$$A_0 := A1 f(a) + A3 f(a) + A2 f(a)$$

$$A_1 := -A3 D(f)(a) + A1 D(f)(a)$$

$$A_2 := \frac{1}{2} A3 (D^{(2)})(f)(a) + \frac{1}{2} A1 (D^{(2)})(f)(a)$$

It is convenient to factor out the derivatives of f in these coefficients

```
> for k from 0 to 2 do B[k]:=simplify(subs(f=(x->x^k/k!),A[k])); od;
```

$$B_0 := A1 + A3 + A2$$

$$B_1 := -A3 + A1$$

$$B_2 := \frac{1}{2}A3 + \frac{1}{2}A1$$

To separate out the first derivative of f we set $B[0]=0$ and $B[1]=1$. We have one more condition available so we set $B[2]=0$.

```
> der1:=solve({B[0]=0,B[1]=1,B[2]=0},{A1,A2,A3});
```

$$der1 := \{A2 = 0, A1 = \frac{1}{2}, A3 = \frac{-1}{2}\}$$

```
> subs(der1,g(h)); taylor(%,h=0,7);
```

$$\frac{1}{2}f(a+h) - \frac{1}{2}f(a-h)$$

$$D(f)(a)h + \frac{1}{6}(D^{(3)})(f)(a)h^3 + \frac{1}{120}(D^{(5)})(f)(a)h^5 + O(h^7)$$

We divide by h to separate out $D(f)(a)$. We see that we will have an error term of order 2. So expanding just through order 2 we obtain the usual form of the central 3-point estimator of the derivative.

```
> subs(der1,g(h))/h: %=taylor(simplify(%,h=0,2);
```

$$\frac{\frac{1}{2}f(a+h) - \frac{1}{2}f(a-h)}{h} = D(f)(a) + O(h^2)$$

We can now define a function which approximates the first derivative

```
> D1B1F1:=unapply(subs(der1,g(h))/h,f,a,h);
```

$$D1B1F1 := (f, a, h) \rightarrow \frac{\frac{1}{2}f(a+h) - \frac{1}{2}f(a-h)}{h}$$

The notation D1B1F1 means first derivative estimate, back 1 step, forward 1 step (so central 3 point).

To get an estimate for the second derivative we proceed in much the same manner:

```
> der2:=solve({B[0]=0,B[1]=0,B[2]=1},{A1,A2,A3});
```

$$der2 := \{A1 = 1, A3 = 1, A2 = -2\}$$

```
> subs(der2,g(h)); taylor(%,h=0,7);
```

$$f(a+h) - 2f(a) + f(a-h) + (D^{(2)})(f)(a)h^2 + \frac{1}{12}(D^{(4)})(f)(a)h^4 + \frac{1}{360}(D^{(6)})(f)(a)h^6 + O(h^7)$$

To separate out the second derivative we divide by h^2 . We see that we will have an error term of order 2. So to get the usual formula we expand the Taylor series (after dividing by h^2) just through order 2.

```
> subs(der2,g(h))/h^2; %=taylor(simplify(%),h=0,2);
```

$$\frac{f(a+h) - 2f(a) + f(a-h)}{h^2} = (D^{(2)})(f)(a) + O(h^2)$$

Here's our formula for the second derivative

```
> D2B1F1:=unapply(subs(der2,g(h))/h^2,f,a,h);
```

$$D2B1F1 := (f, a, h) \rightarrow \frac{f(a+h) - 2f(a) + f(a-h)}{h^2}$$

Let's try now some to find some one-sided 3 point formulae. Note I am redefining g .

```
> g:=h->A1*f(a)+A2*f(a+h)+A3*f(a+2*h);
```

$$g := h \rightarrow A1 f(a) + A2 f(a+h) + A3 f(a+2h)$$

```
> ex:=taylor(g(h),h=0,3);
```

$$ex := (A1 f(a) + A3 f(a) + A2 f(a)) + (A2 D(f)(a) + 2 A3 D(f)(a)) h +$$

$$\left(\frac{1}{2} A2 (D^{(2)})(f)(a) + 2 A3 (D^{(2)})(f)(a) \right) h^2 + O(h^3)$$

```
> for k from 0 to 2 do A[k]:=coeff(ex,h,k); od;
```

$$A_0 := A1 f(a) + A3 f(a) + A2 f(a)$$

$$A_1 := A2 D(f)(a) + 2 A3 D(f)(a)$$

$$A_2 := \frac{1}{2} A2 (D^{(2)})(f)(a) + 2 A3 (D^{(2)})(f)(a)$$

```
> for k from 0 to 2 do B[k]:=simplify(subs(f=(x->x^k/k!),A[k])); od;
```

$$B_0 := A1 + A3 + A2$$

$$B_1 := A2 + 2 A3$$

$$B_2 := \frac{1}{2}A_2 + 2A_3$$

> `der1:=solve({B[0]=0,B[1]=1,B[2]=0},{A1,A2,A3});`

$$\text{der1} := \{A_2 = 2, A_3 = \frac{-1}{2}, A_1 = \frac{-3}{2}\}$$

> `subs(der1,g(h)); taylor(%,h=0,7);`

$$-\frac{3}{2}f(a) + 2f(a+h) - \frac{1}{2}f(a+2h) \\ D(f)(a)h - \frac{1}{3}(D^{(3)}(f)(a))h^3 - \frac{1}{4}(D^{(4)}(f)(a))h^4 - \frac{7}{60}(D^{(5)}(f)(a))h^5 - \frac{1}{24}(D^{(6)}(f)(a))h^6 + O(h^7)$$

> `subs(der1,g(h))/h: %=taylor(simplify(%),h=0,2);`

$$\frac{-\frac{3}{2}f(a) + 2f(a+h) - \frac{1}{2}f(a+2h)}{h} = D(f)(a) + O(h^2)$$

> `D1F2:=unapply(subs(der1,g(h))/h,f,a,h);`

$$D1F2 := (f, a, h) \rightarrow \frac{-\frac{3}{2}f(a) + 2f(a+h) - \frac{1}{2}f(a+2h)}{h}$$

> `der2:=solve({B[0]=0,B[1]=0,B[2]=1},{A1,A2,A3});`

$$\text{der2} := \{A_1 = 1, A_3 = 1, A_2 = -2\}$$

> `subs(der2,g(h)); taylor(%,h=0,7);`

$$f(a) - 2f(a+h) + f(a+2h) \\ (D^{(2)}(f)(a))h^2 + (D^{(3)}(f)(a))h^3 + \frac{7}{12}(D^{(4)}(f)(a))h^4 + \frac{1}{4}(D^{(5)}(f)(a))h^5 + \frac{31}{360}(D^{(6)}(f)(a))h^6 + O(h^7)$$

> `subs(der2,g(h))/h^2; %=taylor(simplify(%),h=0,3);`

$$\frac{f(a) - 2f(a+h) + f(a+2h)}{h^2} \\ \frac{f(a) - 2f(a+h) + f(a+2h)}{h^2} = (D^{(2)}(f)(a)) + O(h)$$

> `D2F2:=unapply(subs(der2,g(h))/h^2,f,a,h);`

$$D2F2 := (f, a, h) \rightarrow \frac{f(a) - 2f(a+h) + f(a+2h)}{h^2}$$

We have obtained 4 estimators:

D1B1F1 = first derivative, 3 point, central, order 2

D2B1F1 = second derivative, 3 point, central, order 2
 D1F2 = first derivative, 3 point, 2 steps forward, order 2
 D2F2 = second derivative, 3 point, 2 steps forward, order 1

The first derivative of the exponential at 0 is 1. Here are our estimates (with step size 0.1 and 0.01):

```
> D1B1F1(exp,0,0.1); D1B1F1(exp,0,0.01);
      1.001667500
      1.000016660
> D1F2(exp,0,0.1); D1F2(exp,0,0.01);
      .9964045700
      .9999664000
```

The second derivative of the exponential at 0 is 1 (of course). Here are our estimates (with step size 0.1 and 0.01):

```
> D2B1F1(exp,0,0.1); D2B1F1(exp,0,0.01);
      1.000833600
      1.000007000
> D2F2(exp,0,0.1); D2F2(exp,0,0.01);
      1.106092200
      1.010060000
```

Note the numerical results support our claims concerning the order.

To obtain higher order derivatives or higher order methods for low order derivatives we have to increase the number of points at which we evaluate f . Here's a central 5 point first derivative estimator:

```
> g:=h->A1*f(a-2*h)+A2*f(a-h)+A3*f(a)+A4*f(a+h)+A5*f(a+2*h);
      g := h → A1 f(a - 2 h) + A2 f(a - h) + A3 f(a) + A4 f(a + h) + A5 f(a + 2 h)
> ex:=taylor(g(h),h=0,5):
> for k from 0 to 4 do A[k]:=coeff(ex,h,k); od:
> for k from 0 to 4 do B[k]:=simplify(subs(f=(x->x^k/k!),A[k]));
od;
```

$$B_0 := A1 + A4 + A2 + A5 + A3$$

$$B_1 := -2 A1 + A4 - A2 + 2 A5$$

$$B_2 := 2 A5 + 2 A1 + \frac{1}{2} A4 + \frac{1}{2} A2$$

$$B_3 := -\frac{1}{6}A_2 + \frac{4}{3}A_5 - \frac{4}{3}A_1 + \frac{1}{6}A_4$$

$$B_4 := \frac{1}{24}A_4 + \frac{1}{24}A_2 + \frac{2}{3}A_5 + \frac{2}{3}A_1$$

> `der1:=solve({B[0]=0,B[1]=1,B[2]=0,B[3]=0,B[4]=0},{A1,A2,A3,A4,A5});`

$$der1 := \{A_3 = 0, A_5 = \frac{-1}{12}, A_1 = \frac{1}{12}, A_2 = \frac{-2}{3}, A_4 = \frac{2}{3}\}$$

> `subs(der1,g(h)); taylor(%,h=0,7);`

$$\frac{1}{12}f(a-2h) - \frac{2}{3}f(a-h) + \frac{2}{3}f(a+h) - \frac{1}{12}f(a+2h)$$

$$D(f)(a)h - \frac{1}{30}(D^{(5)})(f)(a)h^5 + O(h^7)$$

> `subs(der1,g(h))/h: %=taylor(simplify(%),h=0,4);`

$$\frac{\frac{1}{12}f(a-2h) - \frac{2}{3}f(a-h) + \frac{2}{3}f(a+h) - \frac{1}{12}f(a+2h)}{h} = D(f)(a) + O(h^4)$$

> `D1B2F2:=unapply(subs(der1,g(h))/h,f,a,h);`

$$D1B2F2 := (f, a, h) \rightarrow \frac{\frac{1}{12}f(a-2h) - \frac{2}{3}f(a-h) + \frac{2}{3}f(a+h) - \frac{1}{12}f(a+2h)}{h}$$

Note D1B2F2 is a 4 th order method. Let's check it on the exponential:

> `Digits:=30: D1B2F2(exp,0,0.1); D1B2F2(exp,0,0.01); Digits:=10:`

.999996662696097031607111375280

.99999999966662698389550179170

How about a central third derivative estimator?

> `der3:=solve({B[0]=0,B[1]=0,B[2]=0,B[3]=1,B[4]=0},{A1,A2,A3,A4,A5});`

$$der3 := \{A_3 = 0, A_2 = 1, A_4 = -1, A_1 = \frac{-1}{2}, A_5 = \frac{1}{2}\}$$

> `subs(der3,g(h)); taylor(%,h=0,7);`

$$-\frac{1}{2}f(a-2h) + f(a-h) - f(a+h) + \frac{1}{2}f(a+2h)$$

$$(D^{(3)})(f)(a) h^3 + \frac{1}{4}(D^{(5)})(f)(a) h^5 + O(h^7)$$

> `subs(der3,g(h))/h^3: %=taylor(value(%),h=0,4);`

$$\frac{-\frac{1}{2}f(a-2h) + f(a-h) - f(a+h) + \frac{1}{2}f(a+2h)}{h^3} = (D^{(3)})(f)(a) + O(h^2)$$

> `D3B2F2:=unapply(subs(der3,g(h))/h^3,f,a,h);`

$$D3B2F2 := (f, a, h) \rightarrow \frac{-\frac{1}{2}f(a-2h) + f(a-h) - f(a+h) + \frac{1}{2}f(a+2h)}{h^3}$$

Note D3B2F2 is a 2 nd order method. Let's check it on the exponential.

> `Digits:=30: D3B2F2(exp,0,0.1); D3B2F2(exp,0,0.01); Digits:=10:`

1.00250250140593597810947596600
1.00002500025000140542840700000

Here's a 5 point central 4 th derivative estimator:

> `der4:=solve({B[0]=0,B[1]=0,B[2]=0,B[3]=0,B[4]=1},{A1,A2,A3,A4,A5});`

$$der4 := \{A1 = 1, A2 = -4, A4 = -4, A5 = 1, A3 = 6\}$$

> `subs(der4,g(h)); taylor(% ,h=0,7);`

$$\frac{f(a-2h) - 4f(a-h) + 6f(a) - 4f(a+h) + f(a+2h)}{(D^{(4)})(f)(a) h^4 + \frac{1}{6}(D^{(6)})(f)(a) h^6 + O(h^7)}$$

> `subs(der4,g(h))/h^4: %=taylor(simplify(%),h=0,2);`

$$\frac{f(a-2h) - 4f(a-h) + 6f(a) - 4f(a+h) + f(a+2h)}{h^4} = (D^{(4)})(f)(a) + O(h^2)$$

> `D4B2F2:=unapply(subs(der4,g(h))/h^4,f,a,h);`

$$D4B2F2 := (f, a, h) \rightarrow \frac{f(a-2h) - 4f(a-h) + 6f(a) - 4f(a+h) + f(a+2h)}{h^4}$$

Note D4B2F2 is a 2 nd order method. Let's check it on the exponential.

> `Digits:=30: D4B2F2(exp,0,0.1); D4B2F2(exp,0,0.01); Digits:=10:`

1.00166791722900687179959520000
1.00001666679166722883800000000

With 5 points we can for example go back 1 step and forward 3 steps. let's construct such an estimator for the 1 st derivative.

```
> g:=h->A1*f(a-h)+A2*f(a)+A3*f(a+h)+A4*f(a+2*h)+A5*f(a+3*h);
```

$$g := h \rightarrow A1 f(a - h) + A2 f(a) + A3 f(a + h) + A4 f(a + 2 h) + A5 f(a + 3 h)$$

```
> ex:=taylor(g(h),h=0,5):
```

```
> for k from 0 to 4 do A[k]:=coeff(ex,h,k); od;
```

```
> for k from 0 to 4 do B[k]:=simplify(subs(f=(x->x^k/k!),A[k])); od;
```

$$B_0 := A1 + A4 + A2 + A5 + A3$$

$$B_1 := A3 - A1 + 2 A4 + 3 A5$$

$$B_2 := \frac{1}{2} A3 + \frac{1}{2} A1 + 2 A4 + \frac{9}{2} A5$$

$$B_3 := \frac{4}{3} A4 + \frac{1}{6} A3 - \frac{1}{6} A1 + \frac{9}{2} A5$$

$$B_4 := \frac{1}{24} A1 + \frac{2}{3} A4 + \frac{1}{24} A3 + \frac{27}{8} A5$$

```
> der1:=solve({B[0]=0,B[1]=1,B[2]=0,B[3]=0,B[4]=0},{A1,A2,A3,A4,A5});
```

$$der1 := \{A5 = \frac{1}{12}, A3 = \frac{3}{2}, A2 = \frac{-5}{6}, A1 = \frac{-1}{4}, A4 = \frac{-1}{2}\}$$

```
> subs(der1,g(h)); taylor(%,h=0,7);
```

$$-\frac{1}{4} f(a - h) - \frac{5}{6} f(a) + \frac{3}{2} f(a + h) - \frac{1}{2} f(a + 2 h) + \frac{1}{12} f(a + 3 h)$$

$$D(f)(a) h + \frac{1}{20} (D^{(5)})(f)(a) h^5 + \frac{1}{24} (D^{(6)})(f)(a) h^6 + O(h^7)$$

```
> subs(der1,g(h))/h: %=taylor(value(%),h=0,4);
```

$$\frac{-\frac{1}{4} f(a - h) - \frac{5}{6} f(a) + \frac{3}{2} f(a + h) - \frac{1}{2} f(a + 2 h) + \frac{1}{12} f(a + 3 h)}{h} = D(f)(a) + O(h^4)$$

```
> D1B1F3:=unapply(subs(der1,g(h))/h,f,a,h);
```

$$D1B1F3 := (f, a, h) \rightarrow \frac{-\frac{1}{4} f(a - h) - \frac{5}{6} f(a) + \frac{3}{2} f(a + h) - \frac{1}{2} f(a + 2 h) + \frac{1}{12} f(a + 3 h)}{h}$$

D1B1F3 is 4 th order method. Let's check it on the exponential:

```
> Digits:=30: D1B1F3(exp,0,0.1); D1B1F3(exp,0,0.01);  
D1B1F3(exp,0,0.001); Digits:=10:
```

```
1.00000544155730218964608836997
```

```
1.000000000050419058074030602992
```

```
1.000000000000005004169048660750
```

There are many variations! Experiment!

If it is desired to know more about the details of the error we can always calculate a Taylor polynomial. For example for the error in D1B1F3 we have:

```
> taylor( D(f)(a)-D1B1F3(f,a,h),h=0,8);
```

$$-\frac{1}{20}(D^{(5)}(f)(a)h^4 - \frac{1}{24}(D^{(6)}(f)(a)h^5 - \frac{1}{42}(D^{(7)}(f)(a)h^6 + O(h^7)$$

Thus if for example the 5 th derivative of f at a happens to be 0 then D1B1F3(f,a,h) would actually be of 5 th order!

```
>
```